

Storing, analyzing, and presenting Stata output

Julian Reif
University of Chicago
Stata Conference Boston
July 15, 2010

Handling Stata output

- ▶ **Storing**
 - ▶ **regsave, svret**
- ▶ **Analyzing**
 - ▶ Excel PivotTables
- ▶ **Presenting**
 - ▶ **texsave** (LaTeX tables)

Why use these Stata commands?

- ▶ Copy/pasting results is slow and error-prone
- ▶ With these commands, write one Stata script to:
 - ▶ Run analyses
 - ▶ Store and manipulate results
 - ▶ Output results into a table linked to a paper, MS Excel, etc.
- ▶ These commands separate the *storing* of results from the *outputting* of results

Storing output

Storing output

- ▶ Stata commands usually store output in **e()** and **r()**

```
. qui sysuse auto, clear  
. qui summ price  
. return list
```

scalars:

```
      r(N) = 74  
r(sum_w) = 74  
r(mean) = 6165.256756756757  
r(Var) = 8699525.974268789  
r(sd) = 2949.495884768919  
r(min) = 3291  
r(max) = 15906  
r(sum) = 456229
```

- ▶ Modules **regsave** and **svret** convert these into Stata datasets
- ▶ Results can then be manipulated with standard Stata commands

Use **svret** to store returned results

- ▶ **svret** stores returned results
 - ▶ Can save macros and scalars stored in **e()**, **r()**, and **s()**
- ▶ Syntax:

```
svret [classlist] [, long type(type)  
format(%fmt) keep(returnlist)]
```

where

classlist is one or more of the following: *e*, *r*, *s*, *all*,

type can be one of the following: *all*, *scalars*, or *macros*,
and

returnlist is a list of returned results currently in
memory, e.g., **r(N)**.

svret: example 1

```
. sysuse auto, clear  
(1978 Automobile Data)
```

```
. summ price
```

Variable	Obs	Mean	Std. Dev.	Min	Max
price	74	6165.257	2949.496	3291	15906

```
. svret, long format(%8.2fc)
```

```
. list
```

	variable	contents
1.	r(N)	74
2.	r(Var)	8,699,526
3.	r(max)	15,906
4.	r(mean)	6,165.26
5.	r(min)	3,291
6.	r(sd)	2,949.50
7.	r(sum)	456,229
8.	r(sum_w)	74

Use **regsave** to store regression output

- ▶ **regsave** stores regression output
 - ▶ Coefficients, standard errors
 - ▶ Optionally: p-values, t-stats, confidence intervals, covariances, and more
- ▶ User can optionally specify:
 - ▶ Coefficient and variance-covariance matrices
 - ▶ Labels
 - ▶ Table format
- ▶ (Abbreviated) syntax:

```
regsave [coeflist] [using filename] [,  
  tstat pval ci level(#) addlabel(string)  
  addvar(string) table(name, *) *]
```


regsave: example 1

```
. sysuse auto, clear  
(1978 Automobile Data)
```

```
. regress price mpg trunk
```

Source	SS	df	MS			
Model	141126459	2	70563229.4	Number of obs =	74	
Residual	493938937	71	6956886.44	F(2, 71) =	10.14	
Total	635065396	73	8699525.97	Prob > F =	0.0001	
				R-squared =	0.2222	
				Adj R-squared =	0.2003	
				Root MSE =	2637.6	

price	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
mpg	-220.1649	65.59262	-3.36	0.001	-350.9529	-89.3769
trunk	43.55851	88.71884	0.49	0.625	-133.3418	220.4589
_cons	10254.95	2349.084	4.37	0.000	5571.01	14938.89

```
. regsave, tstat pval ci
```

```
. list
```

	var	coef	stderr	tstat	pval	ci_lower	ci_upper	N	r2
1.	mpg	-220.1649	65.59263	-3.356549	.0012707	-350.9529	-89.3769	74	.2222235
2.	trunk	43.55851	88.71884	.4909725	.6249601	-133.3418	220.4588	74	.2222235
3.	_cons	10254.95	2349.084	4.36551	.0000423	5571.01	14938.89	74	.2222235

Analyzing output

Large sets of results can be conveniently analyzed in Excel PivotTables

- ▶ (Instructions for MS Office Excel 2007)
- ▶ Step 1: Open Microsoft Office Excel
- ▶ Step 2: Click Data->From text
 - ▶ Import outsheeted results
 - ▶ Click “Finish”, then “Ok”
- ▶ Step 3: Click Insert->PivotTable, then “Ok”
- ▶ Step 4: Arrange PivotTable as desired

regsave: example 2

```
sysuse auto, clear
tempfile tmpfile
gen lnprice = ln(price)
local replace "replace"

foreach reg in "probit" "logit" "regress" {
    foreach regressors in "lnprice" "lnprice rep78" "lnprice rep78 trunk" {
        foreach stderr in "vce(robust)" "vce(cluster foreign)" "" {

            `reg' foreign `regressors', `stderr'

            regsave using "`tmpfile'", p \\
            addlabel(vce, `stderr', regressors, `regressors', reg, `reg') `replace'

            local replace "append"
        }
    }
}

* Format and outsheet results for use in PivotTable
use "`tmpfile'", clear
replace var = substr(var, "foreign:", "", .)
outsheet using results.txt, replace
```

PivotTable: view 1

	A	B	C
1			
2	vce	vce(robust)	
3	regressors	lnprice rep78 trunk	
4			
5	Values		
6	Row Labels	coef	stderr
7	logit		
8	_cons	-15.110	7.494
9	rep78	1.742	0.483
10	trunk	-0.276	0.088
11	lnprice	1.340	0.870
12	probit		
13	_cons	-8.448	4.246
14	rep78	0.985	0.245
15	trunk	-0.158	0.050
16	lnprice	0.747	0.503
17	regress		
18	_cons	-1.425	0.847
19	rep78	0.246	0.040
20	trunk	-0.040	0.009
21	lnprice	0.168	0.104

PivotTable: view 2

	A	B	C	D	E	F	G
1							
2	vce	vce(robust)					
3	regressors	Inprice rep78 trunk					
4							
5		Column Labels					
6		coef			stderr		
7	Row Labels	logit		probit regress	logit		probit regress
8	_cons	-15.110	-8.448	-1.425	7.494	4.246	0.847
9	rep78	1.742	0.985	0.246	0.483	0.245	0.040
10	trunk	-0.276	-0.158	-0.040	0.088	0.050	0.009
11	Inprice	1.340	0.747	0.168	0.870	0.503	0.104

Presenting output

regsave can create standard tables

```
. qui sysuse auto.dta, clear
.
. qui regress price mpg trunk headroom length
.
. qui regsave using results, table(reg, parentheses(stderr) format(%8.2fc)) replace
.
. qui areg price mpg trunk headroom length, absorb(foreign)
.
. qui regsave using results, table(areg, parentheses(stderr) format(%8.2fc)) append
.
. use results, clear
.
. list
```

	var	reg	areg
1.	mpg_coef	-173.95	-139.90
2.	mpg_stderr	(87.75)	(82.30)
3.	trunk_coef	80.91	38.45
4.	trunk_stderr	(119.82)	(112.24)
5.	headroom_coef	-680.29	-596.43
6.	headroom_stderr	(486.01)	(453.11)
7.	length_coef	23.29	65.43
8.	length_stderr	(27.03)	(28.04)
9.	_cons_coef	6,416.95	-1895.75
10.	_cons_stderr	(6,036.94)	(6,126.44)
11.	N	74	74
12.	r2	0.25	0.36

regsave can create standard tables cont'd

```
. replace var = "" if strpos(var,"stderr")  
(5 real changes made)
```

```
.  
. replace var = substr(var,"_coef","",.)  
(5 real changes made)
```

```
. list
```

	var	reg	areg
1.	mpg	-173.95	-139.90
2.		(87.75)	(82.30)
3.	trunk	80.91	38.45
4.		(119.82)	(112.24)
5.	headroom	-680.29	-596.43
6.		(486.01)	(453.11)
7.	length	23.29	65.43
8.		(27.03)	(28.04)
9.	_cons	6,416.95	-1895.75
10.		(6,036.94)	(6,126.44)
11.	N	74	74
12.	r2	0.25	0.36

Use **texsave** to create LaTeX tables

- ▶ **texsave** creates LaTeX tables
 - ▶ Supports *booktabs* package
 - ▶ 100% compatible with Scientific Workplace
 - ▶ Supports titles, footnotes, borderlines and many LaTeX options
- ▶ (Abbreviated) syntax:

```
texsave [varlist] using filename [if]  
  [in] [, title(string) size(numlist)  
marker(string) hlines(numlist)  
footnote(*) booktabs frag replace *]
```

texsave: example 1

```
. sysuse auto.dta, clear  
(1978 Automobile Data)
```

```
. texsave make mpg trunk if price > 8000 using "table.tex", title(MPG and trunk space) footnote("**Variable trunk is measu  
> red in cubic feet") replace
```

Table 1: MPG and trunk space

make	mpg	trunk
Buick Riviera	16	17
Cad. Deville	14	20
Cad. Eldorado	14	16
Cad. Seville	21	13
Linc. Continental	12	22
Linc. Mark V	12	18
Linc. Versailles	14	15
Olds 98	21	20
Olds Toronado	16	17
Audi 5000	17	15
BMW 320i	25	12
Datsun 810	21	8
Peugeot 604	14	14
Volvo 260	17	14

*Variable trunk is measured in cubic feet

texsave: example 2

```
sysuse auto, clear
tempfile tmpfile
gen lnprice = ln(price)
local replace "replace"
foreach reg in "probit" "logit" "scobit" {

    `reg' foreign lnprice rep78 trunk

    regsave using "`tmpfile'", addlabel(Regression, "`reg'") \\
    table(`reg', asterisk(5 1) parentheses(stderr) format(%8.3fc)) `replace'

    local replace "append"
}
* Create and format LaTeX table
use "`tmpfile'", clear
replace var = substr(var, "foreign:", "", 1)
replace var = substr(var, "_coef", "", 1)
replace var = "" if strpos(var, "stderr")!=0
replace var = "lnalpha" if strpos(var, "lnalpha")!=0
label variable var "Variable name"
local fn "A */** next to coefficient indicates significance at the 5/1\% level."

texsave using "table.tex", title(My regressions) booktabs hlines(10) \\
autonumber footnote("`fn'") varlabels replace
```

texsave: example 2's table

Table 1: My regressions

Variable name	(1)	(2)	(3)
lnprice	0.747 (0.588)	1.340 (1.035)	1.139 (1.793)
rep78	0.985** (0.250)	1.742** (0.479)	3.257 (2.072)
trunk	-0.158** (0.059)	-0.276* (0.108)	-0.493 (0.313)
_cons	-8.448 (4.921)	-15.110 (8.779)	-13.670 (14.637)
lnalpha			-1.281 (0.919)
N	69	69	69
Regression	probit	logit	scobit

A */** next to the coefficient indicates significance at the 5/1% level.

Advanced **texsave** options

- ▶ **frag**
 - ▶ Create a table that can be linked to a LaTeX document via `/input{}` command
- ▶ **marker, align, location**
 - ▶ Give the table a tag and specify its alignment and location
- ▶ **headerlines(string), footlines(string), headlines(string)**
 - ▶ Insert additional LaTeX code before, after, or in the header of your table
- ▶ Many LaTeX table formatting options also available
 - ▶ Alignment and size of footnote
 - ▶ Font styles (italics, bold-faced, etc.)

Feedback and feature requests welcome

- ▶ Email: jreif@uchicago.edu